

# Build and Share Interactive Content with Maple 2025

Maple 2025 provides options for creating documents and interactive applications programmatically. Maple can be used to develop content that can be used interactively within Maple, or automatically translated for use within [Maple Learn](#). Maple Learn is an online environment designed specifically for teaching and learning mathematics.

## Sections

A new [Section](#) element has been added to the [Canvas](#) package. Group content can now contain collapsible sections. These are helpful for organizing content, and can also be used to hiding content that you don't initially want students to see. In this example the Solution section is initially collapsed.

```
1 with (DocumentTools :-Canvas) :
2 cv := NewCanvas ( [
3   Text ("Sketch the graph of: %1" , 3^(1+2*x)),
4   Group ( [
5     Section ( [
6       "Solution" ,
7       StaticPlot (plot (3^(1+2*x)))
8     ] , 'expanded' = false)
9   ] )
10 ] ) :
11 ShowCanvas (cv, entrybox = false);
12
```

Sketch the graph of  $3^{1+2x}$

► Solution

## Check-Box Actions

The [ScriptCheckBox](#) command is used to insert a check box into an interactive application. It can now be given an action that is invoked when the check box is selected or cleared. In this example, the check box has an action handler to show and hide the button immediately when you click it.

```
1 with (DocumentTools :-Canvas ) :
2 showhide := proc ( canvas )
3     local M, sc ;
4     M := GetElements ( canvas , "checkbox" );
5     sc := Script ();
6     SetButton ( sc , "The Button" , "visible" , M[1] :- ischecked );
7     ToString ( sc );
8 end proc :
9
10 cv := NewCanvas ([Text ("Toggle the check box to show and hide the button:" , bold),
11     ScriptButton ("The Button" , "" ),
12     ScriptCheckBox ("The Button is Visible" , showhide , 'ischecked' = true , 'custom' = "cb")
13     ]);
14 ShowCanvas (cv, entrybox = false);
15
```

**Toggle the check box to show and hide the button:**



The Button

The Button is Visible

Additionally, check boxes now have a 'visible' property so they can be shown or hidden as the result of any action.

```
1 with (DocumentTools :-Canvas ):
2 showhide := proc ( canvas )
3     local cb, sc;
4     cb := GetElements (canvas ,"checkbox" )[1];
5     sc := Script ();
6     if cb:-visible = true then
7         SetCheckBox (sc,"cb","visible" ,false);
8         SetButton (sc,"Hide" ,"caption" ,"Show" );
9     else
10        SetCheckBox (sc,"cb" ,"visible" ,true);
11        SetButton (sc,"Show" ,"caption" ,"Hide" );
12    end if;
13
14    ToString (sc);
15 end proc :
16
17 cv := NewCanvas ([Text ("Click the button to show or hide the check box:" ,bold),
18                 ScriptButton ("Hide" , showhide ),
19                 ScriptCheckBox ("The Checkbox" , 'ischecked' = true, 'visible' = true, 'custom' = "cb")
20                 ]):
21 ShowCanvas (cv,entrybox = false);
```




**Click the button to show or hide the check box:**



# Plot Size

Canvas-based plots now accept width and height parameters.

```
1 with(DocumentTools :-Canvas):
2 cv := NewCanvas ( [ StaticPlot ( plot (sin), width = 1000, height = 100 ),
3                     StaticPlot ( plot (cos), width = 200, height = 100 ),
4                     StaticPlot ( plot (tan), width = 200, height = 100 )
5                     ] ):
6 ShowCanvas (cv, entrybox =false);
```




Similarly, the [Quiz](#) command respects the 'plotsize' option when outputting canvas-based quizzes.

```
1 Grading :-Quiz ("Which is a plot of ${FUNC}?" , 1,
2   proc () local A;
3     randomize ();
4     A := [sin(x), cos(x), tan(x)][rand(1..3)()];
5     Grading :-Quiz :-Set ("${FUNC} = A");
6     [plot(A), plot(-A), plot(Pi*A, x = -2*Pi .. 2*Pi)];
7   end proc,
8   gradeisindex = true,
9   title = "Plot Multiple-Choice Randomized" ,
10  style = multiplechoice ,
11  inertform = false,
12  output = showcanvas ,
13  plotsize = [600, 200]
14 );
```

Plot Multiple-Choice Randomized

Which is a plot of  $\sin(x)$ ?



Check Answer

Try Another

